/*

Rochelle Terman	
Proposal title: Human Rights Shaming, Compliance, and Nationalist Backlash

HYPOTHESES

Stated-Hyp1: First, we anticipate that human rights criticism will drive compliance in liberals but backlash in conservatives (H1).
	- NOTE: there is no measure of ideology. we have to use party id instead.

	Test-Hyp1: In liberals, Human rights criticism will worsen perception of human rights conditions in the US

	Test-Hyp2: In liberals, Human rights criticism will increase support for international advocacy

	Test-Hyp3: In liberals, Human rights criticism will increase beliefs about legitimacy of international law

	Test-Hyp4: In liberals, Human rights criticism will increase willingess to take action supporting human rights
	
	
	Test-Hyp5: In conservatives, Human rights criticism will improve perception of human rights conditions in the US
	
	Test-Hyp6: In conservatives, Human rights criticism will decrease support for international advocacy

	Test-Hyp7: In conservatives, Human rights criticism will decrease beliefs about legitimacy of international law

	Test-Hyp8: In conservatives, Human rights criticism will decrease willingess to take action supporting human rights
	
	
Stated-Hyp2: Insofar as it animates national identity, we expect foreign criticism to be less effective than identical criticism emanating from domestic sources (H2). Among democrats, foreign advocacy may still have a net compliance effect, but will be significantly less effective than domestic advocacy. For Republicans, international shaming is likely to compound the backlash effect to an even greater degree. In both groups, the magnitude of the effect should be mediated by nationalist sentiments.

	Test-Hyp9: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism

	Test-Hyp10: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism

	Test-Hyp11: Domestic Human rights criticism will increase beliefs about legitimacy of international law 

	Test-Hyp12: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism

	Test-Hyp13: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 

	Test-Hyp14: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 

	Test-Hyp15: Domestic Human rights criticism will increase beliefs about legitimacy of international law, and this effect will be stronger for Republicans compared to Democrats  

	Test-Hyp16: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 
	
	Test-Hyp17: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism, and this effect will be mediated by nationalist sentiments

	Test-Hyp18: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism, and this effect will be mediated by nationalist sentiments

	Test-Hyp19: Domestic Human rights criticism will increase beliefs about legitimacy of international law, and this effect will be mediated by nationalist sentiments

	Test-Hyp20: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism, and this effect will be mediated by nationalist sentiments
	
Stated-Hyp3: stigmatic shaming is more likely to backfire by affirming citizens’ beliefs that the critic is motivated by intergroup hostility, provoking defensiveness 
	
	Test-Hyp21: Compared to control foreign cricism, stigmatic shaming will improve perception of human rights conditions in the US 

	Test-Hyp22: Compared to control foreign cricism, stigmatic shaming will decrease support for international advocacy compared to foreign criticism

	Test-Hyp23: Compared to control foreign cricism, stigmatic shaming will decrease beliefs about legitimacy of international law 

	Test-Hyp24: Compared to control foreign cricism, stigmatic shaming will decrease willingess to take action supporting human rights compared to foreign criticism
	
	
Stated-Hyp4: Reintegrative shaming mitigates backlash and promotes compliance by lessening the reception of hostile motives

	Test-Hyp25: Compared to control foreign cricism, reintegrative shaming will worsen perception of human rights conditions in the US 

	Test-Hyp26: Compared to control foreign cricism, reintegrative shaming will increase support for international advocacy compared to foreign criticism

	Test-Hyp27: Compared to control foreign cricism, reintegrative shaming will increase beliefs about legitimacy of international law 

	Test-Hyp28: Compared to control foreign cricism, reintegrative shaming will increase willingess to take action supporting human rights compared to foreign criticism
	
********************************************************************************
NOTES
- We used a variable indicating party id (instead of one indicating liberal/
conservative ideology) becasue there is no measure of ideology in the dataset. 
The party id is the closest measure.	
	*/

clear all	
use "Terman1029.dta", clear

********************************************************************************

* INDICATORS OF EXPERIMENTAL MANIPULATION

	* create treatment variable using info from appendix
	lab def treat 1 "control" 2 "foreign critic" 3 "domestic critic" ///
	4 "foreign critic, reintegrative" 5 "foreign critic, stigmatizing"
	gen treat=1 if Q2A==1
	replace treat=2 if Q2B==1
	replace treat=3 if Q2C==1
	replace treat=4 if Q2D==1
	replace treat=5 if Q2E==1
	lab val treat treat
	tab treat, mis
	
	recode treat (1=0) (2/5=1), gen(anycriticism)
	
	* R's party id
	tab PartyID5, mis
	lab def democrat 1 "dem" 0 "rep"
	recode PartyID5 (1/2=1) (4/5=0) (*=.), gen(democrat)
	lab val democrat democrat
	tab democrat, mis
	
	* criticism source: all foreign vs. all domestic
	recode treat (1=.) (2=1) (3=0) (4=1) (5=1), gen(foreign)
	tab foreign, mis
	
	* criticism type: foreign stigmatizing vs. foreign control
	gen stigma=1 if treat==5
	replace stigma=0 if treat==2
	tab stigma, mis                     

	* criticism type: foreign reintegrative vs. foreign control
	gen reintegrate=1 if treat==4
	replace reintegrate=0 if treat==2
	tab reintegrate, mis 
	
* OUTCOME MEASURES (from proposal p. 5)

* perception of domestic human rights condition in the US
	// B.	In general, human rights are well respected in the United States.
	tab Q5_B
	replace Q5_B =. if Q5_B==98
	clonevar humright_respected = Q5_B 
	tab humright_respected, mis
	
	
* support of international advocacy
/*A.I support an international campaign focusing on human rights violations in the United States.
B.	I support a United Nations resolution condemning human rights violations in the United States.
*/

	tab Q9_A
	replace Q9_A=. if Q9_A==98

	tab Q9_B
	replace Q9_B=. if Q9_B==98
	
	gen support_intadvocacy= Q9_A + Q9_B
	tab support_intadvocacy, mis

* belief in legitimacy of international law
/*
A.	Our national government should consistently follow international law. It is wrong to violate international laws, just as it is wrong to violate laws within a country.
B.	If our government thinks it is not in our interest, it should not feel obliged to abide by international laws.
*/

	tab Q10_A
	replace Q10_A=. if Q10_A==98

	tab Q10_B
	replace Q10_B=. if Q10_B==98
	
	gen support_intlaw= Q10_A + Q10_B
	tab support_intlaw, mis
	
* willingness to take action
	* williness to donate to critic org
	recode Q7 (1=1) (2=0) (*=.), gen(willing_donate)
	
	* williness to sign petition
	recode Q8 (1=1) (2=0) (*=.), gen(willing_petition)

	gen willing_act = willing_donate + willing_petition
	tab willing_act, mis
	
* MEDIATOR
* sentiments on national superiority
	// A.	The world would be a better place if other countries were more like the United States.
	tab Q4_A
	replace Q4_A =. if Q4_A==98
	
	clonevar natsuperiority=Q4_A
	tab natsuperiority, mis
	
********************************************************************************

* ANALYSIS

// bonferroni correction needed: 0.05/4
di .05/4 // 0.0125
	
*Test-Hyp1: In liberals, Human rights criticism will worsen perception of human rights conditions in the US

	reg humright_respected i.anycriticism if democrat==1
	// reject.  0.233
	tess 1.anycriticism -, init(Terman1029) bonf(4)	
		
*Test-Hyp2: In liberals, Human rights criticism will increase support for international advocacy

	reg support_intadvocacy i.anycriticism if democrat==1
	// reject.  0.104 
	tess 1.anycriticism +, bonf(4)
	
*Test-Hyp3: In liberals, Human rights criticism will increase beliefs about legitimacy of international law

	reg support_intlaw i.anycriticism if democrat==1
	// reject.  0.479
	tess 1.anycriticism +, bonf(4)
	
*Test-Hyp4: In liberals, Human rights criticism will increase willingess to take action supporting human rights

	reg willing_act i.anycriticism if democrat==1
	// reject.  0.542
	tess 1.anycriticism +, bonf(4)	
	
	
*Test-Hyp5: In conservatives, Human rights criticism will improve perception of human rights conditions in the US
	reg humright_respected i.anycriticism if democrat==0
	// reject.  0.446
	tess 1.anycriticism +, bonf(4)
	
*Test-Hyp6: In conservatives, Human rights criticism will decrease support for international advocacy
	reg support_intadvocacy i.anycriticism if democrat==0
	// reject.  0.019 [test threshold is 0.0125]
	tess 1.anycriticism -, bonf(4)

*Test-Hyp7: In conservatives, Human rights criticism will decrease beliefs about legitimacy of international law
	reg support_intlaw i.anycriticism if democrat==0
	// reject.  0.098 [test threshold is 0.0125]
	tess 1.anycriticism -, bonf(4)
	
*Test-Hyp8: In conservatives, Human rights criticism will decrease willingess to take action supporting human rights
	reg willing_act i.anycriticism if democrat==0
	// reject.  0.803
	tess 1.anycriticism -, bonf(4)

*Test-Hyp9: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism [foriegn will have opposite effect]

	reg humright_respected ib1.foreign
	// reject. 0.507
	tess 0.foreign -, bonf(4)
	
*Test-Hyp10: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism [foriegn will have opposite effect]
	reg support_intadvocacy ib1.foreign
	// reject. 0.685
	tess 0.foreign +, bonf(4)
	
*Test-Hyp11: Domestic Human rights criticism will increase beliefs about legitimacy of international law [foriegn will have opposite effect]
	reg support_intlaw ib1.foreign
	// reject. 0.064
	tess 0.foreign +, bonf(4)
	
*Test-Hyp12: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism [foriegn will have opposite effect]
	reg willing_act ib1.foreign
	// reject. 0.652
	tess 0.foreign +, bonf(4)	

*Test-Hyp13: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 

	reg humright_respected ib1.foreign##ib1.democrat
	// reject. 0.928
	tess 0.foreign#0.democrat -, bonf(4)
	
*Test-Hyp14: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 

	reg support_intadvocacy ib1.foreign##ib1.democrat
	// reject. 0.556
	tess 0.foreign#0.democrat +, bonf(4)
	
*Test-Hyp15: Domestic Human rights criticism will increase beliefs about legitimacy of international law, and this effect will be stronger for Republicans compared to Democrats  

	reg support_intlaw ib1.foreign##ib1.democrat
	// reject. 0.633
	tess 0.foreign#0.democrat +, bonf(4)
	
*Test-Hyp16: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism, and this effect will be stronger for Republicans compared to Democrats 
	
	reg willing_act ib1.foreign##ib1.democrat
	// reject. 0.723
	tess 0.foreign#0.democrat +, bonf(4)
	
*Test-Hyp17: Domestic Human rights criticism will worsen perception of human rights conditions in the US compared to foreign criticism, and this effect will be mediated by nationalist sentiments

	reg humright_respected foreign
	est sto m1

	reg humright_respected foreign natsuperiority
	est sto m2
	
	suest m1 m2
test [m1_mean]foreign=[m2_mean]foreign
// reject. 0.3809

	tess, mediation
	
*Test-Hyp18: Domestic Human rights criticism will increase support for international advocacy compared to foreign criticism, and this effect will be mediated by nationalist sentiments

	reg support_intadvocacy foreign
	est sto m1

	reg support_intadvocacy foreign natsuperiority
	est sto m2
	
	suest m1 m2
test [m1_mean]foreign=[m2_mean]foreign
// reject. 0.3644

	tess, mediation
	
*Test-Hyp19: Domestic Human rights criticism will increase beliefs about legitimacy of international law, and this effect will be mediated by nationalist sentiments

	reg support_intlaw foreign
	est sto m1

	reg support_intlaw foreign natsuperiority
	est sto m2
	
	suest m1 m2
test [m1_mean]foreign=[m2_mean]foreign
// reject. 0.3638

	tess, mediation
	
*Test-Hyp20: Domestic Human rights criticism will increase willingess to take action supporting human rights compared to foreign criticism, and this effect will be mediated by nationalist sentiments
	
	reg willing_act foreign
	est sto m1	
	
	reg willing_act foreign natsuperiority
	est sto m2
	
	suest m1 m2
test [m1_mean]foreign=[m2_mean]foreign
// reject. 0.2930
	
	tess, mediation	
	
*Test-Hyp21: Compared to control foreign cricism, stigmatic shaming will improve perception of human rights conditions in the US 

	reg humright_respected i.stigma
	// reject. 0.033 [test threshold is 0.0125]
	tess 1.stigma +, bonf(4)
	
*Test-Hyp22: Compared to control foreign cricism, stigmatic shaming will decrease support for international advocacy compared to foreign criticism

	reg support_intadvocacy i.stigma
	// reject. 0.170  
	tess 1.stigma -, bonf(4)
	
*Test-Hyp23: Compared to control foreign cricism, stigmatic shaming will decrease beliefs about legitimacy of international law 

	reg support_intlaw i.stigma
	// reject. 0.046
	tess 1.stigma -, bonf(4)
	
*Test-Hyp24: Compared to control foreign cricism, stigmatic shaming will decrease willingess to take action supporting human rights compared to foreign criticism
	
	reg willing_act i.stigma
	// reject. 0.067
	tess 1.stigma -, bonf(4)	


*Test-Hyp25: Compared to control foreign cricism, reintegrative shaming will worsen perception of human rights conditions in the US 

	reg humright_respected i.reintegrate
	// reject. .845
	tess 1.reintegrate -, bonf(4)
	
*Test-Hyp26: Compared to control foreign cricism, reintegrative shaming will increase support for international advocacy compared to foreign criticism

	reg support_intadvocacy i.reintegrate
	// reject. .130
	tess 1.reintegrate +, bonf(4)
	
*Test-Hyp27: Compared to control foreign cricism, reintegrative shaming will increase beliefs about legitimacy of international law 

	reg support_intlaw i.reintegrate
	// reject. .427
	tess 1.reintegrate +, bonf(4)
	
*Test-Hyp28: Compared to control foreign cricism, reintegrative shaming will increase willingess to take action supporting human rights compared to foreign criticism

	reg willing_act i.reintegrate
	// reject. .237
	tess 1.reintegrate +, bonf(4)
